Conditioning Air into the Human Comfort Zone

Introduction

Humans generally feel comfortable between temperatures of 22 °C to 27 °C and a relative humidity of 40% to 60%.

In this application, air at 35 °C and 60% relative humidity will be conditioned into the human comfort zone, with the thermodynamic process plotted on a psychrometric chart. To do this, we will

- first cool the air to 14 °C (this removes some of the water from the air),
- and then heat the air to 24 °C.

Additionally, we will calculate

- the heat and mass of water removed in the cooling phase,
- and the heat added in the heating phase.

Plot the Comfort Zone on the Psychrometric Chart

```
> restart :
  with(ThermophysicalData) : with(plots) :
```

Functions for the lower and upper bounds of the human comfort zone.

```
> lower := T→Property(humidityratio, HumidAir, P = 101325, Tdb = T, R = 0.4) :
```

```
> upper := T→Property(humidityratio, HumidAir, P = 101325, Tdb = T, R = 0.6) :
```

Shade the human comfort zone between 22 °C and 27 °C.

```
> comfortZone := shadebetween(lower, upper, 273.15 + 22 ..273.15 + 27) :
```

Plot the human comfort zone on a psychrometric chart.

```
> display(PsychrometricChart( ), comfortZone)
```



Plotting the Thermodynamic Cycle

Initially the air is at a temperature of 35 °C at a relative humidity of 60%

> $T_1 := 35 + 273.15 :$

> $hr_1 := \text{Property}(\text{humidityratio}, \text{HumidAir}, \text{Tdb} = T_1, \text{"P"} = 101325, \text{"R"} = 0.6)$

$$hr_1 := 0.0215466608092004797$$

Then, we cool the air, and calculate the temperature at saturation (that is, the temperature at which the relative humidity is 1).

> $T_2 := \text{Property}(\text{Tdb}, \text{HumidAir}, R = 1, P = 101325, \text{humidityratio} = hr_1)$

$$T_2 := 299.222762357176293$$

> $hr_2 := hr_1 :$

We continue cooling along the saturation line until we reach 14 °C (in this process, water condenses out of the air).

> $T_3 := 14 + 273.15 :$

> $hr_3 := \text{Property}(\text{humidityratio}, \text{HumidAir}, \text{Tdb} = T_3, P = 101325, R = 1)$

$$hr_3 := 0.0100133227960854491$$

Now we heat the air until it reaches 24 °C.

> $T_4 := 273.15 + 24 :$

> $hr_4 := hr_3$:

Hence the entire thermodynamic cycle can then be plotted.

> route1 := pointplot$\left(\left[\left[T_1, hr_1\right], \left[T_2, hr_2\right]\right]\right.$, connect = true, thickness = 4$\left.\right)$:
route2 := pointplot$\left(\left[\left[T_3, hr_3\right], \left[T_4, hr_4\right]\right]\right.$, connect = true, thickness = 4$\left.\right)$:

> satLine := plot(Property("humidityratio", HumidAir, Tdb = T, P = 101325, R = 1) , T = $T_2 .. T_3$, color = black, thickness = 4) :

> display(PsychrometricChart( ) , comfortZone, route1, route2, satLine)



Heat Changes and Water Removed over the Thermodynamic Cycle

Water removed in the cooling process (in kg water per kg dry air)

> $hr_2 - hr_3$

0.01153333801

Heat removed in the cooling process (in J kg^{-1})

> h_1 := Property(enthalpyperdryair, HumidAir, Tdb = T_1, P = 101325, R = 0.6) :

> h_3 := Property(enthalpyperdryair, HumidAir, Tdb = T_3, P = 101325, R = 1) :

> $h_1 - h_3$

51101.11785

Heat added in the heating process (in J kg^{-1})

> $h_4 := \text{Property}\left(\text{enthalpyperdryair}, \text{HumidAir}, \text{Tdb} = T_4, \text{P} = 101325, \text{R} = 0.5\right) :$

> $h_4 - h_3$

$$8537.89452$$